
PL/SQL reusable code 

Reusable Plsql code 
 
1. INTRODUCTION ..............................................................................................................................2 

2. REUSABLE P_INITIAL_LOAD PROCEDURE .............................................................................2 

 1



PL/SQL reusable code 

Reusable PLSQL code 
 

 
July 2008 

 

1. Introduction 
 

Oracle developers often reproach to PLSQL its non reusability. In my professional experience 
I have successfully loaded a large amount of data into several oracle data bases. During 
these initial loads, I have always used the same stored procedure. This is why I decided to 
share what I believe it represents a reusable code when dealing with oracle initial load. 
 

2. Reusable p_initial_load procedure 
 
During the process of initial load developers are very often faced to the infamous ORA-01555 
rollback segment snapshot too old or to its twin sister ORA-01562 failed to extend rollback 
segment number 10.   
 
I am not going here to discuss how to avoid those two errors but only to precise that the 
reusable code presented here take into account the odds of these errors. So the p_initial_load 
procedure assumes the following facts: 
 

1. The initial load process is restartable 
2. In case of any error in the treatment of one record the error is logged and the next 

record is treated 
3. There is no commit across fetch. Commits are done outside the loop 
4. The execution time of the initial load is assumed to be very long (more than 15 hours) 

this is why an exit from the main loop is done in order to commit outside the loop; 
then the loop is re-opened as from the last treated record. 

 
    
Having said that here below is what I strongly believe can be considered as a reusable code 
when doing initial loads: 
 
CREATE OR REPLACE PROCEDURE P_REUSABLE_INITIAL_LOAD iS 
  
   -- *************************************************************** 
   -- Declare the main functional cursor that will be used 
   -- to initial load data. In general we will receive a flat file 
   -- that we will load in table t_initial_load  via a sql loader 
   -- the very important thing here is to create this table with two 
   -- extra columns named fag_processed (by default equals 'N') and 
   -- error message that will eventually hold the error that might  
   -- occur during the initial load. 
   -- ************************************************************** 
 CURSOR c_main_cursor is 
 SELECT *                 -- '*' should be replaced by the column names 
       FROM t_initial_load 
       WHERE fag_processed = 'N';  -- take only non treated record. It is  
                             -- This where clause which makes the  
      -- Process restartable 
  
  -- *************************************************************** 
  -- This number will be used to exit the technical loop. It will be  
  -- used in order to ensure that all records will be treated 
  -- Remember that we will exit from the loop and will re-enter again 
  -- in order to do not commit across fetch. 
  -- *************************************************************** 
      v_record_number  number; 

 -- 
************************************************************************* 
 -- set the number of record for which we will exit the main loop 
 -- then we will commit and re-enter the loop 
 -- this number can be parameterised in an external table so that we have  
 -- not to change the code if we want to change it. 
 -- I use this commit number only if my initial load is not performant. It  
 -- means that I can't stay a very long time without committing otherwise I  
 -- will get the ORA-01562 error 
****************************************************************************  
     v_commit_val     number := 1000;  
  
     p_error_msg      varchar2(255); 
   
BEGIN 
  
 -- *********************************************************************** 
 -- Start a technical loop which will allow us to exist 
 -- the main functional loop (loop on c_main_cursor), COMMIT 
 -- and re-enter the main loop in order to process the remaining records  
 -- *********************************************************************** 
LOOP 
   
    v_record_number := 0;  
   
 -- *********************************************************************** 
 -- Start the main functional loop 
 -- *********************************************************************** 
  FOR r_main_cursor IN c_main_cursor 
  LOOP 
   
    BEGIN 
     -- log the number of treated record 
        v_record_number := v_record_number + 1; 
 
     -- set a save point here 
   SAVEPOINT  sps_n_savep; 
 
     -- ******************************************************************  
     -- Here do all the treatment you have to do 
     -- it will be better to include these plsql code into a single 
     -- procedure p_treat_record for example. No Commit nor Rollback in this 
     -- procedure 
     -- ****************************************************************** - 
   
       --p_treat_record (r_main_cursor.ide); 
     
     -- ****************************************************************** 
     -- If every thing went well then update the fag_processed of table t 
     -- to 'Y' so that if the process will be restarted it will not treat  
     -- the current record       
     -- ******************************************************************  
   UPDATE t_initial_load t 
   SET   t.fag_processed = 'Y' 
   WHERE t.ide  = r_main_cursor.ide; -- You need to have                     
                                           -- an index on ide 
 
    EXCEPTION 
    WHEN OTHERS THEN 
    -- ************************************************************** 
    -- If an exception occurs in the treatment of the current record 
    -- then we rollback only to save point keeping our successfully 
    -- treated records candidate for a COMMIT 
    -- ************************************************************ 
      ROLLBACK TO sps_n_savep; 
    
    -- **************************************************************** 
    -- We need also to log the error responsible for this 
    -- Unsuccessful record in order to solve it later before 
    -- re-executing the initial load when necessary 
    -- In normal situations you should NEVER USE a WHEN OTHERS 

    -- Exception which is not followed by a RAISE. 
    -- Here it is one of very little number of exceptions to this rule. 
    -- *************************************************************** 
      p_error_msg := substr(SQLERRM,1,255); 
    
      UPDATE t_initial_load t 
      SET    t.error_msg = p_error_msg 
      WHERE  t.ide       = r_main_cursor.ide; 
 
    END; 
    -- ******************************************************* 
    -- Every thing went well for this record. Have we to COMMIT? 
    -- As I wrote above this is to use only if the execution time 
    -- of your initial load is very high 
    -- ******************************************************* 
      IF v_record_number >= v_commit_val THEN 
     EXIT; -- We need then to quit the loop in order to do not commit 
      END IF; -- across fetch which is responsible for the ORA-01555 error 
    
    -- ******************************************************* 
    -- treat the next record  
    -- ******************************************************* 
  END LOOP; -- main functional loop 
   
  -- *********************************************************** 
  -- Here we are outside the main functional Loop, hence 
  -- we can commit successfully treated record 
  -- Of course unsuccessful records have been already rolled back 
  -- to their save points 
  -- *********************************************************** 
    COMMIT; 
  
  -- *********************************************************** 
  -- Now, when we have to finish this initial load? 
  -- When v_record_number is < v_commit_val (1000 in this case) 
  -- This means that we exited the loop while we didn't reach 
  -- the 1000th record. Hence we have no record to be treated 
  -- and we can exit the main technical loop 
  -- *********************************************************** 
    EXIT WHEN v_record_number < v_commit_val ; 
     
END LOOP; -- The main technical Loop 
 
  -- *********************************************************** 
  -- We have finished our initial load. We need just to commit 
  -- the remaining records if not committed yet 
  -- *********************************************************** 
 
    COMMIT; 
    
END P_REUSABLE_INITIAL_LOAD ; 

